

Audacious - OLD, PLEASE USE GITHUB DISCUSSIONS/ISSUES - Bug #271

configure.ac depends on PKG_CHECK_VAR

March 27, 2013 01:52 - Luís Picciochi

Status:	Closed	Start date:	March 27, 2013
Priority:	Major	Due date:	
Assignee:		% Done:	100%
Category:		Estimated time:	0.00 hour
Target version:	3.4		
Affects version:			

Description

The audacious-plugins configure.ac script depends on PKG_CHECK_VAR:

```
$ grep -n PKG_CHECK_VAR configure.ac
1062:PKG_CHECK_VAR([plugindir], [audacious], [plugin_dir], [], [AC_MSG_ERROR([Cannot retrieve plug
in_dir pkgconfig variable]))
```

However, this macro is only available starting at pkg-config 0.28 ([release notes here](#)).

Either increase the dependency on pkg-config (currently at 2.59) or don't use this macro.
As a workaround, I defined the macro locally (ripped from the [upstream commit](#)). I'm not sure if we want to do this even if only when it's not provided by the installed pkg-config.

```
$ git diff configure.ac
diff --git a/configure.ac b/configure.ac
index 0d0f839..db28962 100644
--- a/configure.ac
+++ b/configure.ac
@@ -22,6 +22,16 @@ AUD_COMMON_PROGS
 BUILDSYS_INIT
 BUILDSYS_SHARED_LIB

+ AC_DEFUN([PKG_CHECK_VAR],
+ [AC_REQUIRE([PKG_PROG_PKG_CONFIG])dnl
+ AC_ARG_VAR([$1], [value of $3 for $2, overriding pkg-config])dnl
+ _PKG_CONFIG([$1], [variable="][$3]["], [$2])
+ AS_VAR_COPY([$1], [pkg_cv_] [$1])
+ AS_VAR_IF([$1], [""], [$5], [$4])dnl
+ ])# PKG_CHECK_VAR
+
dnl Headers and functions
dnl =====
AC_CHECK_FUNCS([mkdtemp])
```

History

#1 - March 27, 2013 01:52 - Luís Picciochi

- Target version set to 3.4

#2 - March 27, 2013 03:33 - John Lindgren

Yes, this was also reported at <http://redmine.audacious-media-player.org/boards/1/topics/736>.

#3 - March 27, 2013 16:01 - Michael Schwendt

This would work:

```
m4_ifdef([PKG_CHECK_VAR], [],
[AC_DEFUN([PKG_CHECK_VAR],
  [AC_REQUIRE([PKG_PROG_PKG_CONFIG])dnl
   AC_ARG_VAR([$1], [value of $3 for $2, overriding pkg-config])dnl
   _PKG_CONFIG([$1], [variable="][$3]["], [$2])
   AS_VAR_COPY([$1], [pkg_cv_][$1])
   AS_VAR_IF([$1], [""], [$5], [$4])dnl
])# PKG_CHECK_VAR
])
```

While at it, I just noticed there is also the `AM_PATH_SIDPLAY` macro, which is undefined unless the not-fully-free `libsdlplay2` is installed and provides it. That macro also breaks a simple autoreconf and could be wrapped into `m4_ifdef([AM_PATH_SIDPLAY],[.])`, but that is off-topic here. ;)

#4 - March 30, 2013 01:09 - Lu s Picciochi

I should take more attention when submitting bugs so late in the night. When I said:

Either increase the dependency on `pkg-config` (currently at 2.59) or don't use this macro.

I should have said:

Either declare a dependency on `pkg-config >= 0.28` (somehow, I'm not sure if there's a macro for that) or don't use this macro (in which case, Michael's suggestion above should be good for backwards compatibility).

#5 - March 30, 2013 17:07 - John Lindgren

- Status changed from New to Closed

- % Done changed from 0 to 100

<https://github.com/audacious-media-player/audacious-plugins/commit/f95ab6f939ecf0d9232b3165f9241d2ea9676b9e>